package {
	import br.com.ruidobrasil.manterusuario.UsuarioEdicao;
	import br.com.ruidobrasil.manterusuario.UsuarioSelecaoView;
	import br.com.ruidobrasil.util.TratamentoErros;
	import br.com.ruidobrasil.vo.UsuarioVO;
	
	import flash.events.Event;
	import flash.events.MouseEvent;
	
	import mx.collections.ArrayCollection;
	import mx.controls.Alert;
	import mx.events.CloseEvent;
	import mx.events.FlexEvent;
	import mx.managers.PopUpManager;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;
	
	
	public class UsuarioSelecao extends UsuarioSelecaoView {
		
		private var usuarioEdicao: UsuarioEdicao;
		private var listaUsuarios: ArrayCollection = new ArrayCollection();
		public var idUsuarioLogado: String;
		
		public function UsuarioSelecao() {
			super();
			
			this.addEventListener(FlexEvent.CREATION_COMPLETE, init);
			
			include 'br/com/ruidobrasil/util/RegistroVOs.as';
		}
		
		private function init(event: FlexEvent): void {
			
			btnPesquisar.addEventListener(MouseEvent.CLICK, pesquisar);
			btnIncluir.addEventListener(MouseEvent.CLICK, incluir);
			btnAlterar.addEventListener(MouseEvent.CLICK, alterar);
			btnExcluir.addEventListener(MouseEvent.CLICK, alertaExcluir);
			btnFechar.addEventListener(MouseEvent.CLICK, fechar);
			this.addEventListener(CloseEvent.CLOSE, fechar);
			
			rm.pesquisar.addEventListener(ResultEvent.RESULT, resultPesquisar);
			rm.pesquisar.addEventListener(FaultEvent.FAULT, TratamentoErros.fault);
			rm.excluir.addEventListener(ResultEvent.RESULT, resultExcluir);
			rm.excluir.addEventListener(FaultEvent.FAULT, TratamentoErros.fault);
			
			limparDados();
		}
		
		public function pesquisar(event: MouseEvent = null): void {
			
			var usuario: UsuarioVO = new UsuarioVO();
			usuario.login = txtLogin.actualText == "" ? null : txtLogin.actualText;
			
			rm.pesquisar(usuario);
		}
		
		private function resultPesquisar(event: ResultEvent): void {
			listaUsuarios = event.result.dados.lista;
			grdDados.dataProvider = listaUsuarios;
			
			if (listaUsuarios.length == 0) {
				Alert.show("Nenhum usuário encontrado com o login informado.", "Aviso!");
			}
		}
		
		private function incluir(event: MouseEvent): void {
			usuarioEdicao = new UsuarioEdicao(null);
			usuarioEdicao.telaPai = this;
			PopUpManager.addPopUp(usuarioEdicao, this, true);
			PopUpManager.centerPopUp(usuarioEdicao);
		}
		
		private function alterar(event: MouseEvent): void {
			
			if (grdDados.selectedItem != null) {
				
				usuarioEdicao = new UsuarioEdicao(grdDados.selectedItem as UsuarioVO);
				usuarioEdicao.telaPai = this;
				PopUpManager.addPopUp(usuarioEdicao, this, true);
				PopUpManager.centerPopUp(usuarioEdicao);
				
			} else {
				Alert.show("Selecione um usuário na lista para alterar.", "Aviso!");
			}
		}
		
		private function alertaExcluir(event: MouseEvent): void {
			if (grdDados.selectedItem != null) {
				Alert.yesLabel = "Sim";
                Alert.noLabel = "Não";
				Alert.show("Deseja realmente excluir o usuário selecionado?", "Exclusão", 1|2, this, excluir);
			} else {
				Alert.show("Selecione um usuário na lista e clique em excluir.", "Aviso!");
			}
		}
		
		private function excluir(event: CloseEvent): void {
			if (event.detail == Alert.YES) {
				var usuarioExcluido: UsuarioVO = grdDados.selectedItem as UsuarioVO;
				rm.excluir(usuarioExcluido);
			}
		}
		
		private function resultExcluir(result: ResultEvent): void {
			Alert.show("Usuário excluído com sucesso.", "Aviso!");
			pesquisar();
		}
		
		private function fechar(event: Event): void {
			PopUpManager.removePopUp(this);
		}
		
		private function limparDados(): void {
			txtLogin.actualText = "";
			
			grdDados.dataProvider = null;
		}
	}
}